

# 第1回Linked Open Data (LOD) 活用ワークショップ

2017年8月3日(木)

@理研東京連絡事務所



ハンズオンセッション

LOD用検索言語SPARQLの基礎

大阪大学 産業科学研究所

古崎晃司

## ■ ねらい

- LOD用検索言語SPARQL用いた基本的なクエリの書き方を学ぶ

## ■ 取り組む内容

- 関心があるデータのIRIを探す
- 検索例1: 主語のみ指定
- 検索例2: 主語と述語を指定
- 検索例3: FILTERによる絞り込み
- 検索例4: 述語と目的語を指定
- 検索例5: 複数パターンの組み合わせ
- 検索例6: カウントを利用したランキング

# ハンズオンに必要なデータ



※以下の**2つのURI**は常にWebブラウザで開けるようにしておいて下さい。

## ■ 利用するLODのSPARQLエンドポイント

- 今回のハンズオンでは、  
WikidataのSPARQLエンドポイント(検索用API)  
**<https://query.wikidata.org/>** を利用します。

## ■ 必要なサンプルクエリ, URL等

- 資料置き場  
**<https://github.com/KnowledgeGraphJapan/LODws1st>** にある  
**[/SPARQL-Lec/SPARQL-query-samples.md](#)**  
をご覧ください。

- 1. サンプルクエリの解説
    - スライドで解説＋前のPCでクエリを実行
  - 2. サンプルクエリを各自のPCで試す
    - サンプルクエリをコピー＆ペーストして実行
  - 3. サンプルクエリの一部を，各自の興味に応じて書き換えたクエリ試す
    - サンプルクエリの一部を書き換えて，実行してみる.
    - 試したクエリは，各自の手元のファイルにコピーして残しておく(後のハンズオンでも利用する)
- ※質問があれば，随時，手を挙げてください.

関心があるデータのIRIを探す

# WikidataでのURIの探し方



## 1. Wikidataのページを利用

→各データのページのメニューにある「**Concept URI**」から取得.

※該当ページは, 下記の方法で検索できる

1. 検索欄から「**Wikidata内を検索**」
2. **Wikipedia**の各記事のページのメニューから「**ウィキデータ項目**」のリンクを辿る

## 2. SPARQLクエリを用いた検索

→後ほど説明

## 3. 代表的なURIをまとめたページを利用する

# Wikidataのページ例



Wikidata Query Service x Wikidata Query Service x Osaka University - Wikidata x Osaka University - Wikidata x

保護された通信 | https://www.wikidata.org/wiki/Q651233

**ブラウザで表示されるURLは、このデータのURIとは異なることに注意**

Item Discussion Read View

**Osaka University** (Q651233)

higher education institution in Osaka Prefecture, Japan [edit](#)

▼ In more languages [Configure](#)

Language	Label	Description	Also known as
English	Osaka University	higher education institution in Osaka Prefecture, Japan	
Japanese	大阪大学	No description defined	千里門 豊中キャンパス 大阪医学校 大阪帝国大学 大阪帝大 大阪帝国大学 大阪大学豊中キャンパス ISIR 府立大阪医科大学 吹田キャンパス 阪大 大阪帝国大学 産業科学研究所 大阪大学吹田キャンパス
ꠄucināguci	No label defined	No description defined	
Korean	오사카 대학	No description defined	오사카 대학교

Main page  
Community portal  
Project chat  
Create a new item  
Recent changes  
Random item  
Query Service  
Nearby  
Help  
Donate

Tools  
What links here  
Related changes  
Special pages  
Permanent link  
Page information  
**Concept URI**  
Cite this page

データのURIは、「Concept URI」から取得できる

## 練習: IRIを探して, LODを閲覧



- (1) Wikidata上で「**大阪大学**」の**IRI**を調べ、ブラウザで、リソース(データ)を閲覧する
  - Wikidataの検索欄を利用
  - Wikipediaからのリンクを利用
- (2) 各自が**関心のあるリソースのIRI**を3~5個ほど調べ、ブラウザで閲覧してみる
  - 調べたIRIは、後のクエリで「サンプルの書き換え」に使用するのでメモしておく
  - 固有名詞(インスタンス)を調べることを推奨  
例) 君の名は。 夏目漱石 ドラゴンクエスト
  - データの記述(プロパティの種類)が多いものを探す



# SPARQLクエリの基礎

# SPARQLによるRDFの検索



## ■ SPARQL

- RDFデータに対するクエリ言語
- 「指定したグラフ構造」に一致するトリプルを検索する

## ■ 最も基本的な検索

```
select *  
where {  
  ?s ?p ?o .  
}  
LIMIT 100
```

← 返す要素(\*は全て)

?x (x: 任意の文字列)  
は変数を表す

← 検索するグラフのパターン

「.(ピリオド)」  
を忘れない

↑ 取得する数の制限


このパターンを変  
えることで、欲しい  
データを取得する

この例では  
「任意のトリプルの  
組み合わせ」

# グラフパターンによる検索



## 検索するグラフパターン




## SPARQLでの記述

**?s** **?p** **?o**



# 検索例1:主語のみ指定

検索するグラフパターン




SPARQLでの記述

**<大阪大学> ?p ?o**


# 検索例1: グラフパターン

## 検索するグラフパターン



## SPARQLでの記述

<大阪大学> ?p ?o



# 検索例1: 主語のみ指定



- 「大阪大学」を主語 (**Subject**) に含むトリプルの述語 (**?p**) と目的語 (**?o**) を取得する

```
select *  
where {  
  <http://www.wikidata.org/entity/Q651233> ?p ?o .  
}  
LIMIT 100
```

IRIを直接記述するときは< >で囲む

Wikidataで「大阪大学」を表すURI

## 補足: selectの詳細指定

- ・重複した結果を除く場合  
→ **select distinct** \* とする
- ・取得する変数を明示的に指定する場合  
→ **select ?p ?o** とする

Wikidata Query Service

保護された通信 | <https://query.wikidata.org/#select%20%2a%0Awhere%20%7B%20%0A%20%20%20%3Chttp%3A%2F>

100 Results in 199 ms

Table Download Link

p	o
wikibase:identifiers	15
wikibase:sitelinks	32
wikibase:statements	27
schema:version	504812667
schema:dateModified	Jun 22, 2017
schema:description	universidad japonesa
schema:description	Universität in Japan
schema:description	universiteit in Japan
schema:description	université japonaise
schema:description	universitat japonesa
schema:description	higher education institution in Osaka Prefecture, Japan
schema:description	אוניברסיטה באוסקה, יפן
rdfs:label	جامعة أوساكا
rdfs:label	Osaka Universiteti
rdfs:label	Осакаскі ўніверсітэт
rdfs:label	Universitat d'Osaka
rdfs:label	Osaka University
rdfs:label	Universität Osaka

## Wikidata

<https://query.wikidata.org/>  
での検索結果

## 検索例1-2: 主語のみ指定



- 「大阪大学」を主語 (**Subject**) とするトリプルの述語 (**?p**) を取得する (重複除く)

```
select distinct ?p
where {
  <http://www.wikidata.org/entity/Q651233> ?p ?o .
}
LIMIT 100
```

**Wikidataで「大阪大学」を表すURI**

→このようなクエリを用いることで、  
利用されているプロパティの(種類の)一覧を  
調べることが分かる




## 例1-3: PREFIXを利用した省略表現



- **PREFIX**(接頭語)を定義することで, URIの省略表現が可能

```
select distinct ?p  
where {  
    <http://www.wikidata.org/entity/Q651233> ?p ?o .  
}LIMIT 100
```



```
PREFIX wd: <http://www.wikidata.org/entity/>
```

```
select distinct ?p  
where {  
    wd:Q651233 ?p ?o .  
}LIMIT 100
```

### **PREFIX**の定義

省略表現に用いる文字列は任意に設定できるが, できるだけ慣習的に利用されるものにあわせるとよい.

- (1) 検索例1のクエリを実行する
  - PREFIX利用せず
  - distinctをつける
  - PREFIXを利用
  
- (2) 主語と**IRI**を変え, 同様のクエリ(**PREFIX**利用のみでよい)を実行する
  - 先ほど調べてIRIを使う
  - 各IRIごとに, どのような述語(プロパティ)を持つかを調べる

# 今回の演習で用いるPREFIX



## ■ RDF一般のもの

- **rdfs:** <<http://www.w3.org/2000/01/rdf-schema#>>  
RDFスキーマ(基本的な語彙定義)
- **schema:** <<http://schema.org/>>  
Webのメタデータに記述される語彙
- **skos:** <<http://www.w3.org/2004/02/skos/core#>>  
Web上でのシソーラス, 用語集などに用いられる語彙

## ■ Wikidataで使われるもの

- **wd:** <<http://www.wikidata.org/entity/>>  
エンティティ(もの, コト, データ)
- **wdt:** <<http://www.wikidata.org/prop/direct/>>  
プロパティ(述語/関係)※主にSPARQL検索用の直接関係


## ■ PREFIXの検索サービス

- <https://prefix.cc/>

WikidataのRDFでは, 詳細情報を記述するため, 同じ内容のプロパティが3種類記述されているが, 今回は**wdt:**を使う.

## 検索例2: 主語と述語を指定

検索するグラフパターン




SPARQLでの記述

<大阪大学>  
<本部所在地> ?o

# 検索例②のグラフパターン




## 検索するグラフパターン



## SPARQLでの記述

<大阪大学>  
<本部所在地> ?o



## 検索例2: 主語と述語を指定



### ■ 例2-1)

「大阪大学」の「本部所在地」となる目的語(?o)を取得

```
PREFIX wd: <http://www.wikidata.org/entity/>
PREFIX wdt: <http://www.wikidata.org/prop/direct/>
```

```
select distinct ?o
where {
  wd:Q651233 wdt:P159 ?o .
}LIMIT 100
```

「本部所在地」を表すプロパティ(述語)

## 検索例2: 主語と述語を指定



### ■ 例2-2)

「大阪大学」のラベルとなる目的語(?o)を取得

```
PREFIX wd: <http://www.wikidata.org/entity/>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>

select distinct ?o
where {
  wd:Q651233 rdfs:label ?o .
}LIMIT 100
```

「ラベル」を表すプロパティ(述語)

42 Results in 155 ms

Table Download Link

o	
جامعة أوساكا	
Osaka Universiteti	
Осакскі ўніверсітэт	
Universitat d'Osaka	
Osaka University	
Universität Osaka	
Osaka University	
Universidad de Osaka	
université d'Osaka	
אוניברסיטת אוסקה	
ओसाका विश्वविद्यालय	
Օսական համալսարան	
Universitas Osaka	
Università di Osaka	
大阪大学	
Осака універсітэті	
وساكا وٲندٲو ه مئٲٲهئٲى	
Осака універсітети	
Osaka wniversiteti	
오사카 대학	
Осака Университети	
Universitas Osakensis	



**Wikidata**

<https://query.wikidata.org/>

での検索結果



## 検索例2-3: ラベルの言語種別



- 「大阪大学」のラベルとなる目的語(?o)と、その言語種別を取得

```
PREFIX wd: <http://www.wikidata.org/entity/>  
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
```

```
select distinct ?o  
where {  
  wd:Q651233 rdfs:label ?o .  
}LIMIT 100
```

「ラベル」を表すプロパティ(述語)

「言語種別」を取得する関数

```
select distinct ?o (lang(?o) AS ?ln)
```

とすることで、

「言語の種別」がを合わせて取得できる

結果を別の変数に代入

42 Results in 165 ms

Table Download Link

o	In
جامعة أوساكا	ar
Osaka Universiteti	az
Оса́кскі ўніверсітэт	be
Universitat d'Osaka	ca
Osaka University	cs
Universität Osaka	de
Osaka University	en
Universidad de Osaka	es
université d'Osaka	fr
אוניברסיטת אוסקה	he
ओसाका विश्वविद्यालय	hi
Օսական համալսարան	hy
Universitas Osaka	id
Università di Osaka	it
大阪大学	ja
Осака университеті	kk
وساكا ۋندىۋەرسىتېتى	kk-arab
Осака университеті	kk-cyrl
Osaka wniversiteti	kk-latn
오사카 대학	ko
Осака Университети	ky
Universitas Osakensis	la

**Wikidata**

<https://query.wikidata.org/>

での検索結果

- (1) 検索例2のクエリを実行する
- (2) 主語と**IRI**を変え, 同様のクエリを実行する
  - 検索例2-1の「本部所在地」の代わりとなる, 適当な述語(プロパティ)を選択して用いる

# ヒント: プロパティの探し方



- (1) 適当なデータ(インスタンス)を探し, そのデータの**プロパティ**を調べる
  - データをブラウザで閲覧して調べる
    - Wikidataの場合「**Statements**」欄をみる
  - SPARQLクエリで探す (**検索例1-2**)
    - Wikidataで定義されたプロパティは「**wdt:POO**」のものを使うとよい.  
(同じIDの「**p:POO**」も見つかるが, 今回は使わない)
- (2) プロパティ一覧をまとめたページを調べる
  - **Wikidata**のプロパティ一覧(抜粋)  
[https://www.wikidata.org/wiki/Wikidata:List\\_of\\_properties](https://www.wikidata.org/wiki/Wikidata:List_of_properties)

# Wikidataのデータ閲覧ページ例



Osaka University - Wikidata

保護された通信 | https://www.wikidata.org/wiki/Q651233

### Statements

Property	Value	References
instance of	university	0 references
image	Osaka University Faculty of Medicine.jpg	1 reference
inception	1931	1 reference
name in kana	おおさかだいがく	0 references
country	Japan	0 references
located in the administrative territorial entity	Osaka	

**述語（プロパティ）** (Red box around the left column)

**目的語（オブジェクト）** (Blue box around the right column)

**他のリソースへのリンク** (Yellow box around the "Japan" value)

## ヒント: プロパティを調べるクエリ



### ■ 大阪大学 (wd:Q651233) につながるプロパティを調べる例

#### ■ (1) 大阪大学を主語とするプロパティ

```
PREFIX wd: <http://www.wikidata.org/entity/>  
select distinct ?p  
where { wd:Q651233 ?p ?o . }
```

#### ■ (2) 大阪大学を目的語とするプロパティ

```
PREFIX wd: <http://www.wikidata.org/entity/>  
select distinct ?p  
where { ?s ?p wd:Q651233 . }
```

# 検索例3

## :FILTERによる絞り込み

検索するグラフパターン

任意のグラフパターンと  
組み合わせて使用できる

SPARQLでの記述

**FILTER**(絞り込み条件)

## 例3-1: Filterによる絞り込み



- 「大阪大学」のラベルとなる目的語(?o)から,  
“日本語のラベルのみ”を取得

```
PREFIX wd: <http://www.wikidata.org/entity/>  
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
```

```
select distinct ?o  
where {  
  wd:Q651233 rdfs:label ?o .  
  FILTER (lang(?o) = "ja") .  
}LIMIT 100
```

絞り込みの条件を記述(様々な条件記述できる)



## 例3-2: その他のFilter



- 「大阪大学」のラベルとなる目的語(?o)のうち,  
“Osaka”という文字列を含むものの取得

```
PREFIX wd: <http://www.wikidata.org/entity/>  
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
```

```
select distinct ?o  
where {  
  wd:Q651233 rdfs:label ?o .  
  FILTER (regex(?o, "Osaka")) .  
}LIMIT 100
```

正規表現での  
文字列一致判定

# クエリ練習3




- (1) 検索例3のクエリを実行する
- (2) 先に調べた**IRI**を用いて, 同様のクエリを実行する

## 検索例4

### : 述語と目的語を指定

検索するグラフパターン




SPARQLでの記述

**?s <国> <日本>**


# グラフパターンによる検索

## 検索するグラフパターン



## SPARQLでの記述

?s <国> <日本>



## 検索例4: 述語と目的語を指定



- 例) 4-1: 「国が“日本”と一致する」トリプルの主語(?s)を取得する

```
PREFIX wd: <http://www.wikidata.org/entity/>  
PREFIX wdt: <http://www.wikidata.org/prop/direct/>
```

```
select ?s  
where {  
  ?s wdt:P17 wd:Q17 .  
}  
LIMIT 100
```

日本

国

## 検索例4: 述語と目的語を指定



- 例) 4-2: 「ラベルが“大阪大学”と一致する」トリプルの主語(?s)を取得する

```
PREFIX wd: <http://www.wikidata.org/entity/>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
```

```
select ?s
where {
  ?s rdfs:label "大阪大学"@ja .
}LIMIT 100
```

リテラルの指定

(※データの言語指定も含めて一致が判定される)

★ラベルが分かるとき, IRIを取得できる

「リンクのコピー」  
でURIを取得できる

**Wikidata**

<https://query.wikidata.org/>  
での検索結果

1 Results in 241 ms

s

Q wd:Q651233

# WikidataでのURIの探し方



## 1. Wikidataのページを利用

→各データのページのメニューにある「**Concept URI**」から取得.

※該当ページは, 下記の方法で検索できる

1. 検索欄から「**Wikidata内を検索**」
2. **Wikipedia**の各記事のページのメニューから「**ウィキデータ項目**」のリンクを辿る

## 2. SPARQLクエリを用いた検索

→ラベルの完全一致: **検索例4-2**

→FILTERでラベルの部分一致: **検索例3-2**

## 3. 代表的なURIをまとめたページを利用する

## 検索例4: 主語のクラスを調べる



- 例)4-3)「大阪大学」のクラス(何のインスタンスか?)を取得する


```
PREFIX wdt: <http://www.wikidata.org/prop/direct/>  
PREFIX wd: <http://www.wikidata.org/entity/>
```

```
select ?o  
where {  
  wd:Q651233 wdt:P31 ?o.  
}
```

instance-of

**Wikidata**

<https://query.wikidata.org/>  
での検索結果



university



## 検索例4: 述語と目的語を指定し, 主語の一覧を取得



### ■ 例4-4)「大学」のインスタンスの一覧を取得する

```
PREFIX wdt: <http://www.wikidata.org/prop/direct/>
PREFIX wd: <http://www.wikidata.org/entity/>
select ?s instance-of university
where {
  ?s wdt:P31 wd:Q3918.
}LIMIT 100
```

※例4-3)「wd:Q651233 wdt:P31 ?o」と比較し, 変数を「目的語(?o) から主語(?s)」に変えている  
→この変更により, 「条件を満たす主語(?s)の一覧を取得する」クエリが作れる

- (1) 検索例4のクエリを実行する
- (2) 文字列一致のクエリを用いて, 先に調べたリソースの**IRI**を, **SPARQL**クエリを使って調べることができるか確かめる
- (3) 先に**IRI**を調べたリソースの「クラス」を調べ, 同じクラスのインスタンス一覧を取得するクエリを実行する

## ヒント: クラスの探し方



- (1) 適当なデータ(インスタンス)を探し,  
そのデータの**instance-of**の**目的語**を調べる

- データをブラウザで閲覧
- SPARQLクエリで探す→**検索例4-3**

※ **Wikidata**以外の一般の**LOD**では  
**rdf:type**の**目的語**を調べる


- (2) クラス一覧をまとめたページを調べる
  - **Wikidata**のクラス一覧  
<https://goo.gl/E3zPn8>

# 検索例5

## : 複数パターンの組み合わせ

# グラフパターンの組み合わせ


## 検索するグラフパターン



## SPARQLでの記述

?s ?p1 ?o1.  
?s ?p2 ?o2.


## 検索するグラフパターン



## SPARQLでの記述

?s ?p1 ?o1.  
?o1 ?p2 ?o2.

## 検索するグラフパターン



## SPARQLでの記述

?s ?p1 ?o1.  
?o2 ?p2 ?s.

## 例5-1: 複数の述語を指定して, 目的語を取得する



### ■ 「大阪大学」の“本部所在地”と“創立日”取得する

```
PREFIX wdt: <http://www.wikidata.org/prop/direct/>
```

```
PREFIX wd: <http://www.wikidata.org/entity/>
```

```
select ?o1 ?o2 where {
```

```
  wd:Q651233 wdt:P159 ?o1.
```

```
  wd:Q651233 wdt:P571 ?o2.
```


```
}
```

本部所在地

創立日

クエリを複数書くと,  
**AND**条件として検索  
される

検索するグラフパターン



# 補足: SPARQLの省略表現



## ■ 主語が同じ時の省略表現

```
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
select distinct ?p ?o where {
  ?s rdfs:label "大阪"@ja .
  ?s ?p ?o.
}LIMIT 100
```



```
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
select distinct ?p ?o where {
  ?s rdfs:label "大阪"@ja ;
    ?p ?o.
}LIMIT 100
```

## 例5-2: 述語と目的語を指定し, 主語の一覧を取得(ラベルを併記)




- 「大学」のインスタンスの一覧と,  
その“日本語ラベル付き”を取得する

```
PREFIX wdt: <http://www.wikidata.org/prop/direct/>  
PREFIX wd: <http://www.wikidata.org/entity/>  
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
```

```
select ?s ?o where {  
  ?s wdt:P31 wd:Q3918.  
  ?s rdfs:label ?o .  
  FILTER (lang(?o) = "ja") .  
}LIMIT 100
```

クエリを複数書くと,  
AND条件として検索  
される





100 Results in 1492 ms

Table Download Link

s	o
<a href="#">Q7842</a>	東京大学
<a href="#">Q310842</a>	アル・ムスタンシリーヤ大学
<a href="#">Q312578</a>	フェリックス・メンデルスゾーン・バルトルディ音楽演劇大学ライブツィヒ
<a href="#">Q314089</a>	ミュンヘン専門大学
<a href="#">Q314466</a>	ヘルムシュテット大学
<a href="#">Q13344</a>	ボルドー大学
<a href="#">Q15142</a>	マサチューセッツ大学アマースト校
<a href="#">Q15143</a>	マサチューセッツ大学ローウェル校
<a href="#">Q11942</a>	チューリッヒ工科大学
<a href="#">Q12432</a>	ミシガン工科大学
<a href="#">Q13164</a>	モスクワ大学
<a href="#">Q156598</a>	ライデン大学
<a href="#">Q156725</a>	ハンブルク大学
<a href="#">Q49124</a>	ラドクリフ・カレッジ
<a href="#">Q49167</a>	ウエズリアン大学
<a href="#">Q218506</a>	ルンド大学
<a href="#">Q219317</a>	パヴィア大学
<a href="#">Q219564</a>	カロリンスカ研究所
<a href="#">Q220144</a>	東洋アフリカ研究学院
<a href="#">Q220226</a>	ナミビア大学
<a href="#">Q49204</a>	スミス大学
<a href="#">Q49205</a>	ウェルズリー大学

**Wikidata**

<https://query.wikidata.org/>

での検索結果

## 例5-3: 述語と目的語を指定し, 主語の一覧を取得(ラベルを併記)



- 「大学」のインスタンスの一覧を取得し,  
その“日本語ラベル付き”があれば, 取得する

```
PREFIX wdt: <http://www.wikidata.org/prop/direct/>  
PREFIX wd: <http://www.wikidata.org/entity/>  
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
```

```
select ?s ?o where {  
  ?s wdt:P31 wd:Q3918.  
  OPTIONAL{  
    ?s rdfs:label ?o .  
    FILTER (lang(?o) = "ja") . }  
}LIMIT 100
```

**OPTIONAL{ }**で  
囲うとOR条件となる

100 Results in 870 ms

Table Download Link

s	o
Q wd:Q7842	東京大学
Q wd:Q11942	チューリッヒ工科大学
Q wd:Q12432	ミシガン工科大学
Q wd:Q13164	モスクワ大学
Q wd:Q13334	オルレアン大学
Q wd:Q13335	
Q wd:Q13338	
Q wd:Q13342	
Q wd:Q13340	
Q wd:Q13344	ボルドー大学
Q wd:Q15142	マサチューセッツ大学アマースト校
Q wd:Q15143	マサチューセッツ大学ローウェル校
Q wd:Q15144	
Q wd:Q15574	アデレード大学
Q wd:Q15575	フリンダース大学
Q wd:Q15576	南オーストラリア大学
Q wd:Q16947	同済大学
Q wd:Q16955	清華大学
Q wd:Q16952	北京大学
Q wd:Q22963	シュパイアー行政大学院
Q wd:Q23102	ハルビン工程大学
Q wd:Q23101	北京理工大学

**OR条件**なので、  
「値がない」場合もある

**Wikidata**

<https://query.wikidata.org/>

での検索結果

## クエリ練習5 (a)



- (1) 検索例5-1,5-2,5-3のクエリを実行する
- (2) 例の一部を書き換えたクエリを作って,
  - 「大学」以外のクラスのインスタンス一覧を取得
  - 取得したインスタンス一覧の「日本語ラベル」も併せて取得する

## 例5-4: 主語の述語と, その目的語 の述語を指定



- 「大阪大学」の“本部所在地”と, その“日本語ラベル”取得する

PREFIX wdt: <http://www.wikidata.org/prop/direct/>

PREFIX wd: <http://www.wikidata.org/entity/>

select **?o1 ?o2** where {

wd:Q651233 **wdt:P159** **?o1**.


**?o1** **rdfs:label** **?o2** .

**FILTER (lang(?o2) = "ja") .**

}

本部所在地

検索するグラフパターン



## 例5-5: 主語の述語と, その目的語 の述語を指定



- 「大阪大学」の“クラス”と,  
「大阪大学」の“卒業生”=「大阪大学を“**educated-at**”の目的語とする主語(?s)」を取得する

PREFIX wdt: <http://www.wikidata.org/prop/direct/>

PREFIX wd: <http://www.wikidata.org/entity/>

select **?o1 ?o2** where {

**wd:Q651233** **wdt:P31** **?o1** .


**?o2** **wdt:P69** **wd:Q651233** .

}

**educated-at**

**instance-of**

検索するグラフパターン



## 例5-6: 主語と目的語の組み合わせ

### ■ 「大学」と「卒業生」の組み合わせを取得する

```
PREFIX wdt: <http://www.wikidata.org/prop/direct/>  
PREFIX wd: <http://www.wikidata.org/entity/>  
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
```

```
select ?s ?univ where {  
  ?univ wdt:P31 wd:Q3918.  
  ?s wdt:P69 ?univ.  
}LIMIT 100
```

instance-of

university

educated-at

※例5-5の変数を  
「主語から目的語」に変更

検索するグラフパターン



## 例5-7: 主語と目的語の組み合わせ (日本語ラベル併記)



- 「大学」と「卒業生」の組み合わせを取得する。  
日本語のラベルを併記

```
PREFIX wdt: <http://www.wikidata.org/prop/direct/>
PREFIX wd: <http://www.wikidata.org/entity/>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
select ?univ ?univl ?s ?l where {
  ?univ wdt:P31 wd:Q3918.
  ?s wdt:P69 ?univ.
  OPTIONAL{
    ?univ rdfs:label ?univl .
    FILTER (lang(?univl) = "ja") .
    ?s rdfs:label ?l .
    FILTER (lang(?l) = "ja") . } }LIMIT 100
```



## クエリ練習5 (b)



- (1) 検索例5-4,5-5,5-6のクエリを実行する
- (2) 例の一部を書き換えたクエリを作って,
  - 「大学」と「卒業生」以外の組み合わせを取得
  - 取得したイ組み合わせの「日本語ラベル」も併せて取得する

## 検索例6

### : カウントを利用したランキング

## 検索例6: カウントの利用



### ■ 例6-1) 「大学」のインスタンスの数を取得する

```
PREFIX wdt: <http://www.wikidata.org/prop/direct/>  
PREFIX wd: <http://www.wikidata.org/entity/>  
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
```

データ数をカウントする関数

```
select (count (?s) AS ?c) where {  
  ?s wdt:P31 wd:Q3918.  
}
```

※本来は, **select (count (distinct ?s) AS ?c)**  
とした方が正確な数が得られる.  
(このクエリの場合, 数は変わらず)

# 組み合わせのカウント



## ■ 例6-2) 「大学」と「卒業生の数」の組み合わせをランキング

```
PREFIX wdt: <http://www.wikidata.org/prop/direct/>
PREFIX wd: <http://www.wikidata.org/entity/>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
select ?univ ?univl (count(?s) As ?c) where {
  ?univ wdt:P31 wd:Q3918. ←○○のインスタンス一覧を取得
  ?s wdt:P69 ?univ. ←××のプロパティが「○○のインスタンス」
OPTIONAL{
  ?univ rdfs:label ?univl .
  FILTER (lang(?univl) = "ja") .
}
} GROUP BY ?univ ?univl
ORDER BY DESC(?c)
LIMIT 100
```

データ数を集約(グループ化)する関数

結果を並び変える関数  
※DESCを取ると、小さい順になる

- (1) 検索例6のクエリを実行する
- (2) 検索例の一部を書き換えたクエリを作って、実行する
  - 「大学」と「卒業生」の組み合わせ以外のラインキングを取得するクエリを作成する